Aluno: Guilherme Sales Fernandes
Matrícula: 499756

Arquivo **Main.oz**

```
declare [Circuitos] =
{Module.link['C:/Users/Guilherme/Downloads/Lip_Implementacao2_499756/Circuit.ozf']}

declare A B
A=1|1|_
B=0|1|_ S in
{Circuitos.circuit1 A B S}
{Browse S}

declare X Y Z
X=0|1|_
Y=1|1|_
Z=0|0|_ W in
{Circuitos.circuit2 X Y Z W}
{Browse W}


declare E F G H
E=1|1|_
F=0|0|_
G=0|0|_
H=1|0|_ I in
{Circuitos.circuit3 E F G H I}
{Browse I}
```

Arquivo **Gates.oz**

```
functor
export
    and: AndG
    org: OrG
    xor: XorG
    nand: NandG
    nor: NorG
    notg: NotG
define

fun {GateMaker F}
    fun {$ Xs Ys}
        fun {GateLoop Xs Ys}
            case Xs#Ys of (X|Xr)#(Y|Yr) then
                {F X Y}|{GateLoop Xr Yr}
            end
        end
    in
        thread {GateLoop Xs Ys} end
    end
end

AndG = {GateMaker fun {$ X Y} X*Y end}
OrG = {GateMaker fun {$ X Y} X+Y-X*Y end}
XorG = {GateMaker fun {$ X Y} X+Y-2*X*Y end}
NandG = {GateMaker fun {$ X Y} 1-X*Y end}
NorG = {GateMaker fun {$ X Y} 1-X-Y+X*Y end}

fun {NotG Xs}
    local
        fun {NotLoop Xs}
            case Xs of X|Xr then (1-X)|{NotLoop Xr} end
        end
    in
        thread {NotLoop Xs} end
    end
end

end
```

Arquivo **Circuit.oz**

```
functor
import
    Gates at 'Gates.ozf'
export
    circuit1: PrimeiroCircuito
    circuit2: SegundoCircuito
    circuit3: TerceiroCircuito
define

    proc {PrimeiroCircuito A B ?S}
        E F G H
    in
        E= {Gates.notg A}
        F= {Gates.notg B}
        G= {Gates.and A B}
        H= {Gates.and E F}
        S= {Gates.org G H}
    end

    proc {SegundoCircuito B C D ?S}
        E F G H I
    in
        E= {Gates.notg B}
        F= {Gates.nor B C}
        G= {Gates.xor E F}
        H= {Gates.notg D}
        I= {Gates.nand H F}
        S= {Gates.nand G I}
    end

    proc {TerceiroCircuito A B C D ?S}
        E F G H I J
    in
        E= {Gates.and D C}
        F= {Gates.nand E C}
        G= {Gates.notg B}
        H= {Gates.nor A G}
        I= {Gates.nor A H}
        J= {Gates.org B F}
        S= {Gates.org J I}
    end
end
```

CIRCUITOS IMPLEMENTADOS:

1)



2)



3)